

Un gestionnaire de budget en Calc, sans macro !

Stéphane Quenson

Nous allons ensemble construire un outil de suivi budgétaire avec quelques fonctionnalités avancées comme des dépenses ou recettes régulières et tout cela, sans aucune ligne de code, pour vous montrer la puissance de Calc.

Préparation

Il est important, avant de commencer à bâtir des feuilles de calcul, d'avoir une idée plus ou moins précise de ce que vous souhaitez réaliser. Dans le cas présent, un outil de suivi budgétaire doit permettre de saisir les différentes recettes et dépenses, de les classer par date, de pouvoir facilement créer de nouvelles catégories afin de classer les entrées et sorties d'argent. Il serait aussi utile de pouvoir saisir des détails supplémentaires afin d'analyser plus finement les mouvements. Tout ceci pourrait se résumer en une feuille dont les différentes colonnes seraient : catégorie, date, montant, commentaire.

Une fonctionnalité utile est la saisie des mouvements récurrents, comme la réception mensuelle d'un salaire, un paiement trimestriel pour un club de sport ou un paiement annuel pour un abonnement. Une seconde feuille contiendrait donc dans différentes colonnes la fréquence de paiement (ou de recette !), les dates de début et de fin, le montant, la catégorie et enfin, un commentaire.

Enfin, une troisième feuille montrerait une vue annuelle, par mois et par catégorie avec un total pour chaque ligne et chaque colonne. La liste des catégories serait modifiable et servirait de base pour saisir les recettes et dépenses sur les deux premières feuilles.

Pour terminer l'esthétique, les cellules modifiables auraient un fond bleu ciel et les cellules avec des formules auraient un fond blanc. Cette « astuce » visuelle facilite grandement l'utilisation d'un document, surtout après quelques mois.

Au travail !

Ouvrons un nouveau document Calc et commençons par renommer les trois feuilles en *Détail*, *Récurrent* et *Vue annuelle* en double-cliquant sur chaque nom de feuille. L'ordre des feuilles n'est pas important ! Sauvegardons le docu-




Figure 1. En haut à gauche, le contenu de la « Zone de nom » lorsque la plage A4:A29 est sélectionnée

Gestion du budget familial

ment sous le nom *Budget.ods*. Profitez-en pour lire l'encadré sur les sauvegardes !

La création des catégories se passe dans la feuille *Vue annuelle*. Commençons donc par créer cette information. Sélectionnez les cellules A4:A29, mettez un fond bleu clair et des bordures autour de chaque cellule, nommez cette plage *Catégories* grâce à la « Zone de Nom » (Figure 1) et entrez vos propres catégories, ou les catégories suivantes : Livre, Voiture, Vêtement, Nourriture, Jeux, Essence, Loyer, Restaurant, Salaire, École/Université, Sport, Impôts, Vacances. Ensuite, cliquez D1, mettez le fond bleu, nommez cette plage *Année* puis cliquez sur le menu *Données* -> *Validité* et dans l'onglet *Critère*, choisissez *Autoriser*

Liste puis entrez les années 2009, 2010, ..., 2015 (une année par ligne) et validez. Cette cellule vous permettra de choisir une année et d'afficher les dépenses par mois et par catégorie. Choisissez 2010 comme valeur.

Dans la cellule B3, entrez la formule `=DATE(Année; COLONNE()-1; 1)` et formatez la cellule (menu *Format* -> *Cellule* -> *Nombre*) au format défini par l'utilisateur *MMM-AAAA*. Copiez B3 vers C3:M3. Entrez `"Total "&Année` dans N3 ; centrez les cellules B3:N3 et mettez-les en gras.

Passons maintenant à la feuille *Détail*. Dans les cellules A1:D1, entrez *Catégorie*, *Date*, *Montant* et *Commentaire*, centrez et mettez en gras. Cliquez A2, mettez le fond bleu

et les bordures. Puis cliquez sur le menu *Données* -> *Validité*, choisissez *Autoriser Plage de cellule* et entrez *Catégories* comme nom de plage. Ainsi, seules les catégories créées dans la feuille *Vue Annuelle* pourront être choisies pour entrer des recettes ou dépenses. Copiez la cellule A2 vers A3:A500. Grâce à la « Zone de nom » située juste à gauche du symbole *f(x)* dans la barre de formule, tapez A2:A500, appuyez sur la touche *[Entrée]* pour sélectionner cette plage et tapez *CatDetail* pour nommez cette plage A2:A500.

Ensuite, dans la « Zone de nom », entrez B2:B500, appuyez sur la touche *[Entrée]* et nommez *Date Detail* la plage B2:B500. Mettez le fond bleu et les bordures et formatez comme des dates (JJ/MM/AAAA par exemple).

Ensuite, dans la « Zone de nom », entrez C2:C500, appuyez sur la touche *[Entrée]* et nommez *Montant Detail* la plage C2:C500. Mettez le fond bleu et les bordures et un format `# ##0,00;[RED]-# ##0,00;#` qui affichera les recettes en noir et les dépenses en rouge.

Enfin, dans la « Zone de nom », entrez D2:D500, appuyez sur la touche *[Entrée]* et mettez le fond bleu et les bordures. Il n'y a pas de format spécifique pour la colonne de commentaires.

Saisissez quelques informations de test dans les cellules A2:D4. N'oubliez pas qu'une recette – un salaire par exemple – est un nombre positif alors qu'une dépense est un nombre négatif ! Vérifiez que les dates et montants s'affichent correctement avec le format souhaité (Figure 2).

Revenons maintenant à la feuille *Vue Annuelle*. Dans la cellule B4, entrez la formule : `=SOMMEPROD(CatDetail=$A4; DateDetail>=B$3; DateDetail<C$3; MontantDetail)`.

Cette formule fera la somme des montants relatifs au mois et à la catégorie. Copiez la cellule B4 sur toute la plage B4:L29. Dans la cellule M4, entrez : `=SOMMEPROD(CatDetail=$A4; DateDetail>=M$3; DateDetail<M$3+31; MontantDetail)`, puis copiez cette

	A	B	C	D	E
1					
2		Catégorie	Date	Montant	Commentaire
3	Livres	16-Oct-2006	-85,00		
4	Sport	24-Oct-2006	-87,00		
5	Livres	25-Oct-2006	-25,00		
6	Livres	14-Nov-2006	-656,00		

Figure 2. Quelques entrées de test pour vérifier que les formats sont corrects

	A	B	C	D	E
1					
2		Catégorie	Date	Montant	Message d'erreur
3	Livres	16-Oct-2006	-85,00		Catégorie obsolète!
4	Sport	24-Oct-2006	-87,00		
5	Livres	25-Oct-2006	-25,00		Catégorie obsolète!
6	Livres	14-Nov-2006	-656,00		Catégorie obsolète!

Figure 3. Après le changement du nom de la catégorie en « Livres et BD » dans la feuille *Vue annuelle*, les entrées « Livres », devenues obsolètes, sont clairement signalées

	A	B	C	D	E	L	M
1		ANNEE 2007					
2							
3		Jan-2007	Feb-2007	Mar-2007	Apr-2007	Nov-2007	Dec-2007
4	Livres		-497,00	2.044,00			
30	Catégories obsolètes						
31	Total par mois	-152,00	-649,00	3.347,00	-477,45	-259,00	-275,00

Figure 4. Une vue « condensée » présentant les lignes et colonnes de totaux

	A	B	C	D	E	F	G	H	I	J	K	L
1	Rang	Tous les	Depuis	Jusqu'à	Montant	Catégorie	Message d'erreur					
2	1	2006/11	2007/12	2007/12	300,00	Salaire		1	14	1	14	0
3	2	2006/11	2007/12	2007/12	44,00	Essence		1	14	15	29	28
4	3	2006/11	2007/12	2007/12	222,00	Sport		1	15	16	31	30
5	4							0	0	34	33	32
6	5							0	0	34	33	32
7	6							0	0	34	33	32
8	7							0	0	34	33	32
9	8							0	0	34	33	32
10	9							0	0	34	33	32
11	10							0	0	34	33	32
12	11							0	0	34	33	32
13	12							0	0	34	33	32
14	13							0	0	34	33	32
15	14							0	0	34	33	32
16	15							0	0	34	33	32
17	16							0	0	34	33	32

Figure 5. La construction de la feuille « Récurrent »

cellule vers M5:M29. Sélectionnez la plage B4:M29 et mettez un format # ##0,00;[RED]-# ##0,00;#. Vous devriez voir apparaître les montants de tests.

Changement de catégories

Mais que se passera-t-il si vous changez, dans la feuille *Vue Annuelle*, le catégorie « Livre » en « Livre et BD » ? Les montants correspondants n'apparaîtront plus. Calc n'est pas un outil qui gère facilement les erreurs. Nous vous proposons donc de créer une ligne « Catégories obsolètes » dans la feuille *Vue Annuelle* qui regroupera tous les montants ne pouvant pas être alloués à une catégorie existante. Dans la cellule A30, entrez « Catégories obsolètes » et en B30 entrez la formule : =SOMMEPROD(DateDetail>=B\$3; DateDetail<C\$3; MontantDetail) - SOMME(B4:B29). Copiez B30 vers C30:M30, mettez le format # ##0,00;[RED]-# ##0,00;# et ajoutez des bordures.

De plus, dans la feuille *Détail*, ajoutez un en-tête de colonne en E1 avec le texte *Message d'erreur* et dans E2 tapez : =SI(ET(A2<>""; ESTNA(RECHERCHEV(A2; Catégories; 1; 0))); "Catégorie obsolète !"; "") et copiez cette cellule vers E3:E500. Ainsi, vous identifierez rapidement les entrées à changer (Figure 3). Pour terminer, mettez des bordures autour de E1:E500.

Totaux par mois et catégorie

Dans la feuille *Vue Annuelle*, vous pouvez maintenant entrer les formules suivantes : =SOMME(B4:M4) en N4, puis copiez vers N5:N31. Entrez =SOMME(B4:B30) en B31, puis copiez vers C31:M31. Formatez les plages afin d'afficher les nombres positifs en noir et négatifs en rouge (# ##0,00;[RED]-# ##0,00;#). A31 contiendra le texte « Total par mois ».

Voilà, nous avons déjà un modèle qui fonctionne. Entrez des données dans la feuille *Détail* sur des mois et années différentes, puis revenez sur la feuille *Vue annuelle* et changez d'année en D4. Automati-

Les formules compliquées

Il arrive fréquemment que dans une feuille de calcul, vous soyez amené(e) à écrire des formules très longues. Ma première recommandation est de couper cette formule en plusieurs parties, dans différentes cellules ; l'avantage de cette approche est qu'elle permet de trouver des erreurs plus facilement, en analysant chaque portion de la formule séparément.

Ma seconde recommandation consiste à documenter comment fonctionne votre très longue formule. Sur le moment, lorsque vous l'écrivez, je n'ai aucun doute que vous savez ce que vous faites; malheureusement, après quelques jours ou quelques semaines, relire et comprendre ses propres formules tient parfois du prodige. Donc, ajoutez une note dans la cellule (menu *Insertion* -> *Commentaire*) et décrivez ce que fait votre formule dans un langage clair et précis.

quement, les montants s'ajusteront pour refléter vos recettes et dépenses de l'année choisie. Observez la Figure 4 qui présente les totaux.

Les entrées récurrentes

Dernière étape, facultative, nous allons créer les entrées récurrentes. Dans la feuille *Récurrent*, saisissez les en-têtes suivants dans A1:G1 : *Rang*, *Tous les*, *Depuis*, *Jusqu'à*, *Montant*, *Catégorie* et *Message d'erreur*. Dans A2:A34, mettez les nombres 1, 2, ..., 33. Sélectionnez B2:F34 et mettez le fond bleu. Sélectionnez A1:G34 et créez des bordures tout autour des cellules. Cliquez sur F2 puis sur le menu *Données* -> *Validité*, choisissez *Plage de cellule* et entrez *Catégories* comme nom de plage dans le champ « Source ». Copiez la cellule F2 vers F3:F34. Sélectionnez G2 et tapez =SI(ET(F2<>""; ESTERREUR(RECHERCHEV(F2; Catégories; 1; 0))); "Catégorie obsolète !"; ""); puis copiez G2 dans G3:G34.

Sélectionnez A2:G34 et nommez cette plage *EntréeRécurrente*.

Dans la plage N2:P5, entrons les informations suivantes, ligne par ligne :

Semaine, 0, 7
Mois, 1, 30
Trimestre, 3, 91
Année, 12, 365

Ce petit tableau nous sera bien utile pour calculer les dates exactes des mouvements. Sélectionnez le tableau N2:P5 et nommez-le *Période*.

Cliquez sur B2 puis sur le menu *Données* -> *Validité*. Choisissez *Plage de cellule* et entrez \$N\$2:\$N\$5. Copiez B2 vers B3:B34.

Dans H2, mettez la formule suivante : =SI(OU(B2=""; C2=""; D2=""; D2<C2; E2=""; F2=""); 0; 1). Cette formule vérifie que les paramètres dans les cellules B2:F2 sont corrects et affiche 1 dans ce cas. Si l'une des informations dans B2:F2 est incorrecte, alors H2 affichera 0. Néanmoins, Calc n'est pas fait pour détecter toutes les erreurs, donc attention quand même lorsque vous saisissez des mouvements récurrents.

	A	B	C	D	E	F	G	H	I	J	K	L	M
1													
2													
3		Janv-2010	Fev-2010	Mars-2010	Avr-2010	Mai-2010	Juin-2010	Juil-2010	Août-2010	Sept-2010	Oct-2010	Nov-2010	Déc-2010
4													
5													
6													
7													
8													
9													
10													
11													
12													
13													
14													
15													
16													
17													
18													
19													
20													
21													
22													
23													
24													
25													
26													
27													
28													
29													
30													
31													

Figure 6. La vue annuelle, notez les dépenses récurrentes sur les lignes « Loyer » (paiement mensuel) et « Sport » (paiement trimestriel)

Note importante

Il est important de sauvegarder souvent votre travail, au moins toutes les dix à vingt minutes. Après quelques heures de travail, il est bon de créer une copie du document sous un autre nom, comme ceci par exemple : *<nom du document>_20100930-1345.ods* qui contient la date et l'heure. Une fois par semaine, faites une sauvegarde complète de vos dossiers importants sur une clé USB ou un disque externe. Tout cela ne prend que quelques minutes par mois et sauve souvent plusieurs heures de travail.

En I2, entrez : `=SI(H2; ENT((D2-C2)/RECHERCHEV(B2; Période; 3)+2); 0)`. Cette formule calcule approximativement le nombre d'occurrences à paramétrer pour cette entrée récurrente ; pour ce faire, elle prend la différence entre la date de fin et de début, puis divise par le nombre de jours approximatif pour chaque période. Pour être sûr de ne pas rater la dernière période, nous en ajoutons 2 !

En J2, écrivez `=K1+1`, puis en K1 mettez le chiffre 0 et en K2 tapez `=K1+I2`. Ainsi, la colonne J contiendra la référence de la première entrée de ce mouvement et la colonne K la dernière entrée. Sélectionnez H2:K2 et copiez ce groupe vers H3:K34. Enfin, mettez les nombres 1, ..., 34 dans les cellules L1:L34. Un aperçu du résultat est représenté en Figure 5.

Encore un peu de courage, c'est bientôt fini ! Nous avons maintenant toutes les informations nécessaires pour transformer les mouvements récurrents en entrées individuelles, chacune avec une date correcte d'exécution.

Quelques colonnes de plus...

Sélectionnez maintenant la plage Q2:Q3002 en tapant ces valeurs dans la « Zone de nom », puis cliquez sur le menu *Édition* → *Remplir* → *Série...*, puis tapez 0 comme valeur initiale et 1 comme incrément, cliquez sur *OK* pour terminer, La plage est maintenant remplie des nombres 0 à 3000.

En R2, saisissez `=RECHERCHEV(Q2;K1:L34;2)`. Ainsi, R2 contiendra la référence de l'entrée récurrente. En S2, tapez `=SI(R2=R1;S1+1;0)`, ce qui définira l'occurrence (première période,

seconde période, etc.) pour un mouvement donné. En T2, nous affichons la catégorie par la formule : `=RECHERCHEV(R2;EntréeRécurrente;6)` et en U2, nous affichons simplement la date de début, qui nous servira dans un instant pour calculer la date de chaque occurrence, par `=RECHERCHEV(R2;EntréeRécurrente;3)`.

En V2, pour définir la date de chaque occurrence d'un mouvement, la formule magique est : `=SI(R2>MAX(A1:A34); 0; SI(RECHERCHEV(R2; Entrée Récurrente; 2)="Week"; U2+7*S2; DATE(ANNEE(U2); MOIS(U2)+S2*RECHERCHEV(RECHERCHEV(R2; Entrée Récurrente; 2); Période; 2); JOUR(U2))))`.

Dé-com-po-sons ! Si R2 est supérieur au nombre maximum de mouvements, la formule retourne alors la valeur 0 qui correspond au 31-Dec-1899. Sinon, le mouvement est-il hebdomadaire ? Dans ce cas, le calcul de chaque date est relativement simple : à partir de la date de départ, nous ajoutons 7 jours par période (`U2+7*S2`). Si le mouvement est mensuel ou trimestriel, nous ajoutons avec la fonction *DATE* le nombre correct de mois à la date de départ. C'est le moment de lire mes recommandations au sujet des formules compliquées !

Enfin, en W2, mettons le montant de chaque entrée : `=SI(ESTERREUR(V2); 0; SI(V2>RECHERCHEV(R2; Entrée Récurrente; 4); 0; RECHERCHEV(R2; EntréeRécurrente; 5)))`.

Il ne nous reste plus qu'à copier la plage R2:W2 vers R3:W3002, ainsi, les entrées récurrentes seront proprement calculées. Pour vérifier que tout fonctionne correctement,

entrez dans la plage B2:F2 les valeurs suivantes : Mois, 01-Jan-2010, 01-Dec-2010, -500, Loyer, et vérifiez que les cellules R2:W14 contiennent effectivement les dépenses mensuelles de -500, sauf W14 qui doit contenir un 0 car la date en V14 (01-Jan-2011) est supérieure à la date de fin de cette entrée récurrente (01-Dec-2010).

Si tout est correct, la dernière étape consiste à modifier la formule qui résume les recettes et dépenses dans la feuille *Vue annuelle*. Dans cette feuille, cliquez sur B4 et ajoutez à la fin de la formule : `+SOMMEPROD(Récurrent.T2:T3002=$A4; Récurrent.$V$2:$V$3002>=B$3; Récurrent.V2:V3002<C$3; Récurrent.$W$2:$W$3002)`. Copiez cette cellule vers B4:L29. Cliquez ensuite sur M4 et ajoutez à la fin de la formule : `+SOMMEPROD(Récurrent.T2:T3002=$A4; Récurrent.$V$2:$V$3002>=M$3; Récurrent.V2:V3002<M$3+31; Récurrent.$W$2:$W$3002)`. Copiez ensuite vers M5:M29.

Il ne reste plus qu'à modifier la formule pour les catégories obsolètes en ajoutant ceci à la formule de B30 : `+SOMMEPROD(Récurrent.V2:V3002>=B$3; Récurrent.$V$2:$V$3002<C$3; Récurrent.W2:W3002)`. Copiez ensuite B30 vers C30:L30, puis ajoutez à la formule de M30 : `+SOMMEPROD(Récurrent.V2:V3002>=M$3; Récurrent.$V$2:$V$3002<M$3+31; Récurrent.W2:W3002)`.

Conclusion

Comme vous l'avez constaté, construire une feuille de calcul sophistiqué ne prend que quelques dizaines de minutes en combinant des fonctions puissantes comme *SOMMEPROD* ou *RECHERCHEV*.

La Figure 6 représente la vue annuelle qui inclut des paiements récurrents. Il n'est pas toujours nécessaire de recourir aux macros mais pour être tout à fait franc, elles permettent d'aller encore plus loin et de gérer au mieux les erreurs, fournissant ainsi une meilleure qualité des données. ■